Extended Gibbs Sampler for 1-type BGW

In this script:

a0 = death probability

a1 = inactivity probability

a2 = division probability

functions

```
(*create conditional likelihoods*)
CondLikelihood[j_, k_] := D[(a0 + a1 * s + a2 * s^2)^j, {s, k}] / k! /.
  s → 0  (*j_ = start with j cells, k_=end up with k cells*)
```

```
In[∘]:= (*The GibbsSampler function takes the likelihood as
 input and the number of points to draw from the distribution*)
GibbsSampler[likelihood_, points_] :=
 (
  pa0 = ConstantArray[0., points];
  pa1 = ConstantArray[0., points]; pa2 = ConstantArray[0., points];
  (*starting point: the algorithm can also start from
     a generic point found from a initial Dirichlet distribution*)
  {a0Cur , a1Cur, a2Cur } = {0.2, 0.5, 0.3};
  possibleMonomials = MonomialList[likelihood];
   (*The likelihood is split into several monomials,each of which represents
   a tree that connects two cell counts at distinct generations.*)
  exponents = Exponent[#, {a0, a1, a2}] & /@ possibleMonomials ;
  For[l = 1, l ≤ points, l++,
    (*found the probabilities to select
     a particular tree given the current parameter values*)
   weights = possibleMonomials /. {a0 → SetPrecision[a0Cur , 100],
      a1 → SetPrecision[a1Cur, 100], a2 → SetPrecision[a2Cur, 100] };
   weights  = weights / Total[weights];
    (*Print["weights]",weights];*)

   multinomial = RandomChoice[weights → exponents];
    (*sample a tree given the found weights*)

    (*sample from the updated Dirichlet*)
   {a0Cur, a1Cur} = RandomVariate[DirichletDistribution[multinomial + 1]];
    (*p(\vec{\alpha}|\{Z_k(n)\}^{(l)})*)
    a2Cur = 1 - a0Cur - a1Cur;
   pa0⟦l⟧ = a0Cur; pa1⟦l⟧ = a1Cur; pa2⟦l⟧ = a2Cur;
  ]
 )
```

Results : likelihood P $\left(Z_{n+1} = 78 \mid Z_n = 40, \vec{\alpha}\right)$

Calculate the analytical posterior and marginals

In[ ]:=
```
likelihood = N[CondLikelihood[40, 78]]
likelihood = likelihood /. a0 → 1 - a1 - a2; (*constraint*)
(*Print["likelihood:", Expand[likelihood]]*)
prior = 2.;
posterior = prior *
   likelihood / Integrate[Integrate[prior * likelihood, {a2, 0, 1 - a1}], {a1, 0, 1}];
Plot3D[If[a1 + a2 ≤ 1, posterior], {a1, 0, 1}, {a2, 0, 1},
 PlotRange → {All, All, All}, ColorFunction → "BlueGreenYellow",
 ImageSize → 200, PlotLegends → posterior , AxesLabel → {a1, a2}]
```

Out[ ]=

$8.83058 \times 10^{-116} \left(8.83294 \times 10^{117}\, a1^2\, a2^{38} + 4.52971 \times 10^{116}\, a0\, a2^{39}\right)$

Out[ ]=



$7.49254 \times 10^{-113} \left(4.52971 \times 10^{116}\, a2^{39}\, (-a1 - a2 + 1) + 8.83294 \times 10^{11}\right.$

In[ ]:=
```
(*symbolic marginals*)
pa1 = Integrate[posterior, {a2, 0, 1 - a1}];
pa2 = Integrate[posterior, {a1, 0, 1 - a2}];
posterior = posterior /. a2 → z - a1 ;
pz = Integrate[posterior, {a1, 0, z}]; pa0 = pz /. z → 1 - a0 ;
```

In[ ]:=
```
ylim = 24;
plt0 = Plot[pa0, {a0, 0.000001, 1.}, AxesLabel → {a0},
   PlotStyle → Directive[Opacity[0.6], Red], PlotRange → {{0, 1}, {0, ylim }}];
plt1 = Plot[pa1, {a1, 0.000001, 1.}, AxesLabel → {a1},
   PlotStyle → Directive[Opacity[0.5], Green], PlotRange → {{0, 1}, {0, ylim }}];
plt2 = Plot[pa2, {a2, 0, 1.0}, AxesLabel → {a2},
   PlotStyle → Directive[Opacity[0.3], Blue], PlotRange → {{0, 1}, {0, ylim }}];
{plt0, plt1, plt2}
```

Out[ ]=



Calculate the numerical marginals

In[]:=
```
points = 5000;
likelihood = CondLikelihood[40, 78];
GibbsSampler[likelihood, points];
```

In[]:=
```
(*trace plots*)
tracea0 = ListLinePlot[pa0, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
tracea1 = ListLinePlot[pa1, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
tracea2 = ListLinePlot[pa2, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
{tracea0, tracea1, tracea2}
```

Out[]=



In[]:=
```
nbin = 35;
hista0 = Histogram[pa0, nbin, "ProbabilityDensity",
   AxesLabel → {a0}, PlotRange → {{0., 1}, {0, 20}}, ChartStyle → EdgeForm[]];
hista1 = Histogram[pa1, nbin, "ProbabilityDensity",
   AxesLabel → {a1}, PlotRange → {{0., 1}, {0, 25}}, ChartStyle → EdgeForm[]];
hista2 = Histogram[pa2, nbin, "ProbabilityDensity",
   AxesLabel → {a2}, PlotRange → {{0., 1}, {0, 15}}, ChartStyle → EdgeForm[]];
{Show[hista0, plt0], Show[hista1, plt1], Show[hista2, plt2]}
```

Out[]=



$1, \vec{\alpha}$

Results : likelihood P $\left(Z_{n+3} = 3 \mid Z_n = 1, \vec{\alpha}\right)$

```
(* calculate the likelihood*)
pgf = a0 + a1 * s + a2 * s^2;
gen = 3;
maxcell = 2^gen;
likelihoods = Table[0, 2^gen + 1];
For[i = 1, i ≤ gen - 1, i++, pgf = pgf /. s → pgf;];
For[k = 0, k ≤ maxcell, k++,
  likelihood = D[pgf, {s, k}] / k! /. s → 0; likelihoods[[k + 1]] = likelihood];
```

In[ ]:=
```
likelihood = likelihoods⟦4⟧; (*observing 3 cells after 3 generations *)
likelihood = likelihood /. a0 → 1 - a1 - a2; (*constraint*)
(*Print["likelihood:", Expand[likelihood]]*)
prior = 2;
posterior = prior *
   likelihood / Integrate[Integrate[prior * likelihood, {a2, 0, 1 - a1}], {a1, 0, 1}];
Plot3D[If[a1 + a2 ≤ 1, posterior], {a1, 0, 1}, {a2, 0, 1},
 PlotRange → {All, All, All}, ColorFunction → "BlueGreenYellow",
 ImageSize → 200, PlotLegends → posterior , AxesLabel → {a1, a2}]
```

Out[ ]=



$$\frac{3063060\left(a1\left(12\,a1^2\,a2^2+a2\left(24\,a1\,a2^2\left(a2\left(-a1-a2+1\right)^2+a1\left(-a1-a2+1\right)-a1-a2+1\right)+6\left(2\,a1\,a2\right.\right.\right.\right.}{}$$

In[ ]:=
```
(*symbolic marginals*)
pa1 = Integrate[posterior, {a2, 0, 1 - a1}];
pa2 = Integrate[posterior, {a1, 0, 1 - a2}];
posterior = posterior /. a2 → z - a1;
pz = Integrate[posterior, {a1, 0, z}]; pa0 = pz /. z → 1 - a0;
```

In[ ]:=
```
ylim = 6;
plt0 = Plot[pa0, {a0, 0.000001, 1.}, AxesLabel → {a0},
   PlotStyle → Directive[Opacity[0.6], Red], PlotRange → {{0, 1}, {0, ylim }}];
plt1 = Plot[pa1, {a1, 0.000001, 1.}, AxesLabel → {a1},
   PlotStyle → Directive[Opacity[0.5], Green], PlotRange → {{0, 1}, {0, ylim }}];
plt2 = Plot[pa2, {a2, 0, 1.0}, AxesLabel → {a2},
   PlotStyle → Directive[Opacity[0.3], Blue], PlotRange → {{0, 1}, {0, ylim }}];
{plt0, plt1, plt2}
```

Out[ ]=



Use the Gibbs Sampler algorithm

In[ ]:=
```
points = 20000;
likelihood = likelihoods⟦4⟧; (*observing 3 cells after 3 generations *)
GibbsSampler[likelihood, points]
```

In[∘]:=
```
(*trace plots*)
tracea0 = ListLinePlot[pa0, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
tracea1 = ListLinePlot[pa1, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
tracea2 = ListLinePlot[pa2, Frame → True, FrameTicks → {{0, 50000, 10^5}, Automatic}];
{tracea0, tracea1, tracea2}
```

Out[∘]=



In[∘]:=
```
nbin = 35;
hista0 = Histogram[pa0, nbin, "ProbabilityDensity",
   AxesLabel → {a0}, PlotRange → {{0., 1}, {0, 4}}, ChartStyle → EdgeForm[]];
hista1 = Histogram[pa1, nbin, "ProbabilityDensity",
   AxesLabel → {a1}, PlotRange → {{0., 1}, {0, 4}}, ChartStyle → EdgeForm[]];
hista2 = Histogram[pa2, nbin, "ProbabilityDensity",
   AxesLabel → {a2}, PlotRange → {{0., 1}, {0, 4}}, ChartStyle → EdgeForm[]];
{Show[hista0, plt0], Show[hista1, plt1], Show[hista2, plt2]}
```

Out[∘]=